

УДК 004.942.519.87(045)

О. Г. Додонов, А. І. Кузьмичов

Інститут проблем реєстрації інформації НАН України
вул. М. Шпака, 2, 03113, Київ, Україна
e-mail: dodonov@ipri.kiev.ua; akuzmychov@gmail.com

Аналіз ризиків в організації, плануванні, виконанні та супроводженні НДДКР стандартними засобами імітаційного моделювання Excel

Характерна специфіка НДДКР¹ — вимушене врахування умов невизначеності та ризиків на усіх етапах виконання робіт, у першу чергу, щодо ресурсних складових проекту/програми: фінансових, матеріальних, кадрових тощо. Тому й не дивно, що інструментальні засоби аналітичного апарату імітаційного моделювання виникли², розробляються і активно використовуються як неодмінна складова цієї сфери. Вказані умови суттєво ускладнюють процеси формування, обговорення і прийняття організаційних/управлінських рішень, бо зазвичай сфера НДДКР стосується надзвичайно відповідальних проектів, що здійснюються в жорстких фінансових і часових межах, швидкий і наближений розв'язок задачі є чи не єдиним прийнятним варіантом, який можна отримати в лабораторному практикумі ВНЗ, користуючись стандартними засобами Excel.

Ключові слова: імітаційне моделювання, ризик-аналіз, метод Монте-Карло, simulation modeling.

Вступ

Світова практика свідчить про масовий характер використання інструментальних засобів імітаційного моделювання (ІМ), зокрема, в підготовці кваліфікованих управлінців, це досить складні комерційні програмні продукти двох основних

¹ Науково-дослідні та дослідно-конструкторські роботи (сукупність робіт, спрямованих на отримання нових знань та їхнє практичне застосування при створенні нового виробу або технології), англ. Research & Development, R&D. НДР: роботи пошукового, теоретичного та експериментального характеру, що виконуються з метою визначення технічної можливості створення нової техніки в певні терміни. ДКР: комплекс робіт з розробки конструкторської і технологічної документації на дослідний зразок виробу, виготовлення та випробувань дослідного зразка виробу.


² Методи Монте-Карло, метод PERT в управлінні проектами, гібридна технологія Simulation Optimization.

класів: пакети спеціального призначення із застосуванням специфічних мов, що розраховані на програмістів, і ліцензовані програми-надбудови для популярних електронних таблиць (ЕТ). Щодо останніх користувачі ЕТ, у першу чергу, Excel (студенти, аспіранти, викладачі) мають змогу користуватися лише разовими free-версіями цих надбудов і, зрозуміло, з вимушеними обмеженнями на розмір задачі (число невизначених змінних, кількість спроб ГВЧ) і період користування (15÷30 діб).

Оскільки популярна науково-навчальна література з проблематики ризик-аналізу засобами ІМ зазвичай використовує певну надбудову³ Excel [2–5], здається, що це єдиний шлях до побудови та реалізації імітаційної моделі.

У цій роботі пропонується підхід до побудови ІМ стандартними засобами Excel, який дозволяє ознайомитися з моделями цього типу і, за потреби, забезпечити регулярну навчально-дослідницьку роботу без придбання та довантаження додаткових програмних продуктів.

Традиційно в аналітичній практиці діє принцип «Що-Якщо» (What-If Analysis), за яким кожна дія (Якщо, входи) неодмінно супроводжується наслідком (Що, виходи), ідеально його реалізацією є електронні таблиці за модельним підходом «входи-виходи» шляхом побудови сценаріїв. У Excel є набір інструментів цього аналізу, його складові для ІМ: *Диспетчер сценаріїв* і *Таблиця даних*.



Сценарний підхід зводиться до побудови набору сценаріїв, від найгіршого до найкращого, де за заданими фіксованими значеннями входів автоматично обчислюються значення виходів. Це можна зробити за допомогою надбудови *Диспетчер сценаріїв*, визначивши порівняно невелику кількість сценаріїв; недолік: важко отримати узагальнену оцінку виходів за певним спектром значень входів.

Універсальною та кращою для практики сценарного підходу є *симуляція*⁴ — побудова специфічної комп'ютерної моделі із застосуванням методів Монте-Карло (статистичних випробувань), де здійснюється багатокрокова (у сотнях і тисячах) процедура імітації поведінки досліджуваного об'єкта/системи, де кожен крок — окремий сценарій. Кінцевим результатом є не конкретні значення виходів, а їхні узагальнені статистичні оцінки у вигляді, скажімо, очікуваної тривалості проекту зі значенням $1234,56 \pm 78,9$ днів, і відповідного розподілу.

³ Основних три: @RISK (Palisade), Crystal Ball (Oracle) та, найчастіше, Analytic Solver (Frontline Systems).


⁴ *Симуляція* (лат. *simulatio*) — відтворення ключових властивостей обраної реальної або абстрактної системи, її проводять з метою: тренування та навчання персоналу, тестування технології у граничних/нештатних умовах, розваг і ситуаційних/штабних ігор, синоніми: *імітація* або, як метод дослідження, *моделювання*. На практиці це експериментування, яке проводять на математичній, фізичній (симуляторі) чи гібридній моделі, коли робити це в реальній системі/об'єкті недосяжно, дорого чи небезпечно (згадаймо експерименти на об'єкті ЧАЕС). З появою ЕОМ і комп'ютерного моделювання поняття *simulation* стало скороченою формою напрямку *computer simulation*, де застосовують спеціальні мови типу *Simula* та ГВЧ (генератори випадкових чисел). Оскільки тоді в СРСР симуляція сприймалася лише як негативна поведінка людини-симулянта, розглядалася з позицій медичної (зображення хвороби або окремих її симптомів людиною, яка не страждає на дане захворювання) або юридичної практики (переслідування корисливих цілей: ухилення від призову або несення обов'язків військової служби чи отримання незаслужених матеріальних піл'єг), з тих пір напрям *simulation modeling* або просто *simulation* у публікаціях (укр., рос.) отримав назву та використовується у перекладах не інакше, як *імітаційне моделювання*.

Методи Монте-Карло — група чисельних методів для дослідження випадкових процесів на основі базового алгоритму, де:

- 1) входи: задано множини визначених (констант) і невизначених (із визначеними розподілами ймовірностей значеннями) початкових даних;
- 2) для кожного заданого розподілу ймовірностей треба побудувати ГВЧ (генератор випадкових чисел) і налаштувати його для відповідної невизначеної змінної;
- 3) виходи: запуском сукупності ГВЧ і обчисленнями за відповідними формулами отримують статистичні оцінки шуканих змінних і, якщо можливо, їхні розподіли.

Інструментальні засоби Excel для імітаційного моделювання

Для ІМ стандартними засобами Excel є: інструменти надбудови Data Analysis (рос. Анализ данных); статистичні функції; інструменти надбудови Аналіз «якщо»; засоби для роботи з таблицями великого розміру.



Інструменти надбудови *Data Analysis* (рос. Анализ данных) для ІМ — це комплект із 7 ГВЧ популярних розподілів, багатообіцяючих, але, насправді, непридатних для роботи — згенерований ними набір даних *статичний*, це аналог давнішніх паперових таблиць випадкових чисел разового використання, ознака — клавіша F9 не оновлює щойно створений набір, тож єдиним стандартним засобом Excel для організації ІМ залишаються статистичні функції.

Головною серед них є функція RAND() (в інтерфейсі рос. СЛЧИС()): це ГВЧ пристойної для навчання якості⁵ з рівномірним розподілом у діапазоні значень [0, 1). Важливо, що ця функція застосовується для побудови ГВЧ інших розподілів, скажімо, нормального для універсального застосування чи трикутного — для моделювання проектів.

У надбудові Аналіз «якщо» корисним є інструмент *Таблиця даних*, за допомогою якого можна зручно побудувати таблиці великого розміру для генерування значень невизначених вхідних змінних і обчислення значень виходів (без потреби прокрутки і перегляду сотень рядків).

⁵ Якість ГВЧ — довжина циклу генерування до чергового повтору, для RAND() це 2^{31} (≈ 2 млрд.) значень функції, для ІМ вважається «короткою»: ГВЧ з «довгими» циклами у спеціальних надбудовах мають оцінку 2^{200} і більше.

Характерна властивість ІМ в Excel — використання таблиць досить великого розміру із сотнями рядків, де без витрат часу на прокрутку треба отримати результати.

Приклади

Приклад 1. Обчислювальна математика.

Методом Монте-Карло, тобто, утворенням вибірки входів генеруванням випадкових чисел заданого розподілу, обчислити площу кола з центром (0,5; 0,5) і, відповідно, радіусом 0,5, вписаного в квадрат з одиничними сторонами, за тимчасового припущення, що число π невідоме.

Таким наближеним, але економним обчислювальним методом (чисельного інтегрування) для складних задач визначають, наприклад, площі складних поверхонь типу водоймищ або ж об'єми опуклих n -вимірних тіл, границі яких важко описати аналітично.

Ідея: функцією RAND() генеруються по 1000 значень координат точок (x, y), які як краплинки дощу падають на квадрат, шукана приблизна площа круга — це кількість точок у крузі/1000, для яких відстань d від центру не перевищує 0,5, вимога до ГВЧ — рівномірність, а точність — чим більше точок, тим вища.

Використані стандартні інструменти Excel: виділення областей, *Прогресія* (нумерація) та *Таблиця даних* (заповнення діапазону розміром 1000×4 значеннями формул), завдяки яким не треба прокручувати діапазони великих розмірів.

Реалізація наведеного прикладу проілюстрована на рис. 1, 2.

1	№	X	Y	d	0/1	S	S _T
2						0,7790	0,785398
3	1	0,184922046	0,75570319	0,405781021	1		
4	2	0,968152039	0,03888672	0,65710866			
5	3	0,573245241	0,97724826	0,482836174	1		
6	4	0,534723564	0,40390179	0,102179212	1		
7	5	0,38900839	0,21169835	0,308928762	1		
998	996	0,966419736	0,72422124	0,517515733			
999	997	0,048581417	0,3481596	0,476271186	1		
1000	998	0,688089653	0,5836021	0,199202012	1		
1001	999	0,617497029	0,85462426	0,373582548	1		
1002	1000	0,157626551	0,35979123	0,369970375	1		

	A	B	C	D	E	F	G	H
1	№	X	Y	d	0/1		S	S _T
2							=SUM(E3:E1002)/1000	=PI()*0,5^2
3	1	=RAND()	=RAND()	=SQRT((B3-0,5)^2+(C3-0,5)^2)	=IF(D3<=0,5;1;"")			
4	2	=TABLE(A2)	=TABLE(A2)	=TABLE(A2)	=TABLE(A2)			
5	3	=TABLE(A2)	=TABLE(A2)	=TABLE(A2)	=TABLE(A2)			
6	4	=TABLE(A2)	=TABLE(A2)	=TABLE(A2)	=TABLE(A2)			
7	5	=TABLE(A2)	=TABLE(A2)	=TABLE(A2)	=TABLE(A2)			
998	996	=TABLE(A2)	=TABLE(A2)	=TABLE(A2)	=TABLE(A2)			
999	997	=TABLE(A2)	=TABLE(A2)	=TABLE(A2)	=TABLE(A2)			
1000	998	=TABLE(A2)	=TABLE(A2)	=TABLE(A2)	=TABLE(A2)			
1001	999	=TABLE(A2)	=TABLE(A2)	=TABLE(A2)	=TABLE(A2)			
1002	1000	=TABLE(A2)	=TABLE(A2)	=TABLE(A2)	=TABLE(A2)			

Рис. 1. Робоча таблиця

Результат.

Наближене значення площі кола $\approx 0,7790$, точне (πr^2): 0,785..., тобто, прогнозом 1000 спроб функції RAND() для кожного входу гарантується вірний перший знак після десяткової коми (7) та близький до «8» другий знак ($S \approx 0,78$).




Рис. 2. Точкова діаграма (x, y), частотна діаграма довжини (точка-центр)⁶

Приклад 2. Бізнес-аналітика.

Виробничий підрозділ науково-дослідної організації за результатами виконаних НДДКР має скласти бізнес-план пробного випуску і реалізації протягом року партії розробленого конкурентоспроможного виробу. За цим планом очікуваний прибуток визначається за заданими значеннями 5 вхідних змінних, і усі вони невизначені, адже організація, явно ризикуючи, вперше виходить на ринок як продавець, без серйозних маркетингових досліджень, і тому попередньо формує імітаційну модель для «програвання» поведінки ринкового середовища.

Цими невизначеними змінними (входами) є питомі оцінки для окремого виробу, залежні від розміру партії: ціна, оплата праці та закупівлі, а також загальні витрати та попит для усієї партії. За їхніми значеннями очікуваний прибуток (вихід) обчислюється за формулою:

$$\text{Прибуток} = \text{Попит} \times (\text{Ціна} - \text{Оплата праці} - \text{Закупівлі}) - \text{Загальні витрати.}$$

За сценарним підходом для орієнтації будуть пару сценаріїв, найгірший і найкращий, де за заданими значеннями входів автоматично обчислюється значення виходу.

1. Метод Монте-Карло: властивості входів.

Невизначеність входів задана різними розподілами:

- 1) ціна — трикутний;
- 2) загальні витрати — цілочисловий рівномірний;
- 3) оплата праці — дискретний;
- 4) закупівлі — рівномірний;
- 5) попит — нормальний.

2. Реалізація.

За допомогою базової функції RAND() шляхом допоміжних обчислень налаштовують необхідний ГВЧ трикутного розподілу.


Входи: три оцінки: a — оптимістична, b — песимістична та m — «нормальна, скоріше за все».

Функцію трикутного розподілу та її реалізацію показано на рис. 3–5:

⁶ Підтвердження ЦГТ (центральної граничної теореми) про нормальність функції виходу при будь-яких розподілах входів.

$$f(a, m, b) = \begin{cases} a + \sqrt{R(b-a)(m-a)}, & R \leq F \\ b - \sqrt{(1-R)(b-a)(b-m)}, & R > F \end{cases}$$

$$R = \text{RAND}(), F = \frac{m-a}{b-a}$$



3. Входи.

Параметри		Питомі	Ціна	
Ціна		379,9	a	230
Заг. витрати		81522	m	400
Оплата праці		250,0	b	550
Закупівлі		67,29	R	0,4130
			F	0,5313

А	В
1 Параметри	Питомі
2 Ціна	=IF(H5<=H6;H2+SQRT(H5*(H4-H2)*(H3-H2));H4-SQRT((1-H5)*(H4-H2)*(H4-H3)))
3 Заг. витрати	=INT(G11*RAND()+G12)
4 Оплата праці	=VLOOKUP(RAND();A11:C15;3;TRUE)
5 Закупівлі	=F11+(F12-F11)*RAND()
6	
7 Попит	=INT(NORM.INV(RAND();F14;F15))
8 Прибуток	=B7*(B2-B4-B5)-B3

Оплата праці				Закупівлі	Заг. витрати
Нижня	Верхня	Питома варт.	Ймовірн.	Нижня границя	Верхня границя
0	0,1	130	0,1	60	20000
0,1	0,3	150	0,2	120	80000
0,3	0,7	250	0,4	Попит	
0,7	0,9	280	0,2	Середнє 15000	
0,9	1	300	0,1	Ст. відх. 4500	

Рис. 3. Налаштування входів

Формування вибірки (5 входів, 1 вихід, по 1000 значень).

ГВЧ. Таблиця даних						
Спроба	Ціна	Заг. витрати	Оплата праці	Закупівлі	Попит	Прибуток
1	361,9808604	80440	250	81,09	10670	249154,3
2	379,8410159	92486	150	93,56	8845	1112919,1
3	296,5574691	95275	250	111,04	16601	-1165824,1
4	333,3879862	93583	150	64,34	18824	2147407,9
5	336,3614003	84601	280	105,11	22811	-1196644,0
996	313,94729	97387	150	92,99	13980	894643,3
997	294,5355252	85367	250	108,96	16938	-1176629,2
998	248,0662093	81226	250	117,36	10747	-1363236,6
999	249,7803076	95635	280	92,55	6975	-951919,9
1000	314,2766547	91988	250	61,05	5197	-75208,6

Рис. 4. Вибірка (фрагмент)

Результат (один із можливих)




Рис. 5. Нормальний розподіл

За заданими параметрами невизначених входів отримано оцінки очікуваного прибутку, де ризик зриву бізнес-плану складає біля 20 %, розподіл виходу близький до нормального.

Приклад 3. Проектний менеджмент.

Для роботи з унікальними проектами, для яких ще нема історії чи будь-якого досвіду, а лише припущення щодо тривалості робіт, у 1958 р. було розроблено метод PERT (Program Evaluation and Review Technique) для мережевої моделі проекту, який зазвичай реалізують в Excel із застосуванням надбудови Analytic Solver [3, 9]. На відміну від детермінованого методу СРМ (Critical Path Method), методом PERT відшуковують не фіксоване значення очікуваної тривалості проекту (інакше, довжини критичного шляху, КШ), а оцінки нормального розподілу значення КШ (середнє та дисперсію).

Покажемо, як за наведеною тут методикою можна реалізувати цей метод стандартними засобами Excel шляхом побудови формульної моделі без дотримання будь-яких умов щодо розмірів задачі.


Підхід: для кожної роботи визначається розподіл тривалості та параметри, формується таблиця спроб, за якої обчислюються статистичні оцінки тривалостей робіт та усього проекту. У прикладі на початку використовується рівномірний

розподіл тривалості в діапазоні (a, b) , що реалізується формулою $a + (b - a) \text{RAND}()$, який надалі можна замінити трикутним (a, m, b) .

Початкові дані: 11 робіт-вузлів ($r0 \div r10$), 18 дуг-зв'язків, робоча таблиця, де:

1) згенеровані параметри і тривалості робіт;

2) обчислені часові параметри проекту: РС, РФ, ПС, ПФ, ЗР (ранні та пізні старти/фініші, загальний резерв, довжина КШ) за формульною моделлю (без застосування оптимізатора) демонструє рис. 6.



	A	B	C	D	E	F	G	H	I	J	K
5	Номер	Робота	Попередн.	a	b	t	РС	РФ	ПС	ПФ	ЗР
6	1	r0		1	13	7,31	0,00	7,31	0,00	7,31	0,00
7	2	r1	r0	1	9	6,97	7,31	14,28	16,06	23,02	8,70
8	3	r2	r0	5	13	7,23	7,31	14,54	14,88	22,11	7,60
9	4	r3	r0	4	15	9,00	7,31	16,32	7,31	16,32	0,00
10	5	r4	r1	5	13	12,46	14,28	26,74	23,02	35,48	8,70
11	6	r5	r3	2	13	5,79	16,32	22,11	16,32	22,11	0,00
12	7	r6	r1, r2, r3	5	13	11,47	16,32	27,79	24,01	35,48	7,70
13	8	r7	r4, r6	5	15	7,60	27,79	35,38	35,48	43,08	7,70
14	9	r8	r2, r5	5	19	15,77	22,11	37,87	22,11	37,87	0,00
15	10	r9	r3, r6, r8	4	12	5,20	37,87	43,08	37,87	43,08	0,00
16	11	r10	r6, r7, r9	1	11	3,48	43,08	46,56	43,08	46,56	0,00
17				КШ=		46,56					

	A	B	C	D	E	F	G
5	Номер	Робота	Попередн.	a	b	t	РС
6	1	r0		=RANDBETWEEN(1;5)	=D6+RANDBETWEEN(6;14)	=D6+(E6-D6)*RAND()	=MAX(IF(ISERROR(SEARCH(\$B\$6:\$B\$16;C6));0;\$H\$6:\$H\$16))
7	2	r1	r0	=RANDBETWEEN(1;5)	=D7+RANDBETWEEN(6;14)	=D7+(E7-D7)*RAND()	=MAX(IF(ISERROR(SEARCH(\$B\$6:\$B\$16;C7));0;\$H\$6:\$H\$16))
8	3	r2	r0	=RANDBETWEEN(1;5)	=D8+RANDBETWEEN(6;14)	=D8+(E8-D8)*RAND()	=MAX(IF(ISERROR(SEARCH(\$B\$6:\$B\$16;C8));0;\$H\$6:\$H\$16))
9	4	r3	r0	=RANDBETWEEN(1;5)	=D9+RANDBETWEEN(6;14)	=D9+(E9-D9)*RAND()	=MAX(IF(ISERROR(SEARCH(\$B\$6:\$B\$16;C9));0;\$H\$6:\$H\$16))
10	5	r4	r1	=RANDBETWEEN(1;5)	=D10+RANDBETWEEN(6;14)	=D10+(E10-D10)*RAND()	=MAX(IF(ISERROR(SEARCH(\$B\$6:\$B\$16;C10));0;\$H\$6:\$H\$16))
11	6	r5	r3	=RANDBETWEEN(1;5)	=D11+RANDBETWEEN(6;14)	=D11+(E11-D11)*RAND()	=MAX(IF(ISERROR(SEARCH(\$B\$6:\$B\$16;C11));0;\$H\$6:\$H\$16))
12	7	r6	r1, r2, r3	=RANDBETWEEN(1;5)	=D12+RANDBETWEEN(6;14)	=D12+(E12-D12)*RAND()	=MAX(IF(ISERROR(SEARCH(\$B\$6:\$B\$16;C12));0;\$H\$6:\$H\$16))
13	8	r7	r4, r6	=RANDBETWEEN(1;5)	=D13+RANDBETWEEN(6;14)	=D13+(E13-D13)*RAND()	=MAX(IF(ISERROR(SEARCH(\$B\$6:\$B\$16;C13));0;\$H\$6:\$H\$16))
14	9	r8	r2, r5	=RANDBETWEEN(1;5)	=D14+RANDBETWEEN(6;14)	=D14+(E14-D14)*RAND()	=MAX(IF(ISERROR(SEARCH(\$B\$6:\$B\$16;C14));0;\$H\$6:\$H\$16))
15	10	r9	r3, r6, r8	=RANDBETWEEN(1;5)	=D15+RANDBETWEEN(6;14)	=D15+(E15-D15)*RAND()	=MAX(IF(ISERROR(SEARCH(\$B\$6:\$B\$16;C15));0;\$H\$6:\$H\$16))
16	11	r10	r6, r7, r9	=RANDBETWEEN(1;5)	=D16+RANDBETWEEN(6;14)	=D16+(E16-D16)*RAND()	=MAX(IF(ISERROR(SEARCH(\$B\$6:\$B\$16;C16));0;\$H\$6:\$H\$16))
17				КШ=		=H16	

H	I	J	K
РФ	ПС	ПФ	ЗР
=G6+F6	=J6-F6	=MIN(IF(ISERROR(SEARCH(B6;C6:C16));MAX(H6:H16);I6:I16))	=ROUND(I6-G6;1)
=G7+F7	=J7-F7	=MIN(IF(ISERROR(SEARCH(B7;C7:C17));MAX(H7:H17);I7:I17))	=ROUND(I7-G7;1)
=G8+F8	=J8-F8	=MIN(IF(ISERROR(SEARCH(B8;C8:C18));MAX(H8:H18);I8:I18))	=ROUND(I8-G8;1)
=G9+F9	=J9-F9	=MIN(IF(ISERROR(SEARCH(B9;C9:C19));MAX(H9:H19);I9:I19))	=ROUND(I9-G9;1)
=G10+F10	=J10-F10	=MIN(IF(ISERROR(SEARCH(B10;C10:C20));MAX(H10:H20);I10:I20))	=ROUND(I10-G10;1)
=G11+F11	=J11-F11	=MIN(IF(ISERROR(SEARCH(B11;C11:C21));MAX(H11:H21);I11:I21))	=ROUND(I11-G11;1)
=G12+F12	=J12-F12	=MIN(IF(ISERROR(SEARCH(B12;C12:C23));MAX(H12:H23);I12:I23))	=ROUND(I12-G12;1)
=G13+F13	=J13-F13	=MIN(IF(ISERROR(SEARCH(B13;C13:C24));MAX(H13:H24);I13:I24))	=ROUND(I13-G13;1)
=G14+F14	=J14-F14	=MIN(IF(ISERROR(SEARCH(B14;C14:C25));MAX(H14:H25);I14:I25))	=ROUND(I14-G14;1)
=G15+F15	=J15-F15	=MIN(IF(ISERROR(SEARCH(B15;C15:C26));MAX(H15:H26);I15:I26))	=ROUND(I15-G15;1)
=G16+F16	=J16-F16	=MIN(IF(ISERROR(SEARCH(B16;C16:C27));MAX(H16:H27);I16:I27))	=ROUND(I16-G16;1)

Рис. 6. Проектна мережа, робоча таблиця

Генерування ВЧ (по 300 спроб), обчислення статистичних оцінок 11 робіт і проекту, інфографіка, результат (рис. 7).

▲	A	B	C	D	E	F	G	H	I	J	K	L	M	▲	A	N	O	P	Q	R	S	T	U	V	W	X
18													7,64	18		Належність критичному шляху										
19													stdev	19		100%	17%	11%	72%	7%	70%	22%	22%	72%	79%	100%
20	Середн.	7,78	7,92	8,22	8,38	8,33	8,47	8,78	8,46	9,02	8,17	8,01	50,35	20	Середн.	300	51	34	217	22	211	66	66	215	238	300
21	Спроба	r0	r1	r2	r3	r4	r5	r6	r7	r8	r9	r10	КШ	21	Спроба	r0	r1	r2	r3	r4	r5	r6	r7	r8	r9	r10
23	1	8,1	9,3	3,0	14,6	6,9	4,2	2,2	6,5	10,8	16,6	5,9	60,1	23	1	0	16,80	13,7	0,0	20,2	0,0	9,2	17,3	0,0	0,0	0,0
24	2	5,3	11,0	7,5	11,3	10,3	7,9	3,4	4,9	5,8	7,0	9,1	46,4	24	2	0	12,8	3,5	0,0	16,3	0,0	4,2	7,9	0,0	0,0	0,0
25	3	4,3	7,7	5,9	8,7	5,4	6,2	8,9	8,9	14,2	10,7	5,9	50,1	25	3	0	2,5	4,7	0,0	2,5	0,0	13,5	2,5	0,0	0,0	0,0
26	4	15,7	8,9	12,1	7,5	11,4	7,8	13,7	17,1	12,1	7,2	6,5	65,1	26	4	0	8,5	10,1	0,0	8,8	0,0	8,5	8,5	0,0	0,0	0,0
27	5	7,7	5,8	16,6	2,8	7,0	6,0	6,1	4,7	2,9	11,4	7,9	49,8	27	5	0	26,0	14,5	0,0	27,3	0,0	20,6	22,1	0,0	0,0	0,0
318	296	3,6	4,8	6,4	13,5	11,7	5,2	6,3	14,4	5,0	10,0	2,6	40,3	318	296	0	5,9	12,5	0,0	12,1	0,0	5,9	11,6	0,0	0,0	0,0
319	297	8,4	1,1	12,1	7,0	9,3	4,6	13,6	11,9	7,3	2,1	11,0	57,1	319	297	0	7,4	7,3	0,0	9,3	0,0	4,5	4,5	0,0	0,0	0,0
320	298	4,4	1,4	10,3	6,7	4,8	8,2	7,5	9,9	8,5	5,1	7,0	39,8	320	298	0	9,3	13,1	0,0	9,3	0,0	11,3	9,3	0,0	0,0	0,0
321	299	8,5	14,6	12,9	8,0	7,0	4,8	8,0	15,9	9,3	11,0	9,7	56,8	321	299	0	14,5	13,1	0,0	19,1	0,0	6,7	6,7	0,0	0,0	0,0
322	300	6,6	11,7	10,5	3,7	4,5	2,1	9,1	9,0	4,6	6,4	11,9	48,3	322	300	0	11,9	6,2	0,0	20,7	0,0	11,9	20,7	0,0	0,0	0,0




Рис. 7. Результати ІМ

Результат: критичний шлях: тривалість (поточний сеанс) $T_{кр} = 50,35 \pm 7,64$, розподіл близький до нормального.

Приклад 4. Організаційне управління.

До розробок, названих проривними [10], відносять гібридну технологію *simulation-optimization* (SO), де спільно і ефективно діють два класи моделей, імітаційні та оптимізаційні, які за традицією зазвичай розглядаються окремо. SO-технологія нарешті задовольняє компромісну умову невизначеності входів і визначеності значення цільових функцій в оптимізаційних задачах — симулятор формує область значень невизначених змінних за їхніми розподілами, де оптимізатор здійснює оптимальний вибір на цій основі.

Зрозуміло, що розв'язок в Excel оптимізаційних задач у такий спосіб зводиться до побудови досить громіздкої таблиці з формулами, значення яких регулярно змінюються і відповідною обробкою поступають до модуля оптимізації, але познайомитися з цією технологією варто, хоча зазвичай це предмет спеціальних надбудов.

Типовою задачею є оптимальний вибір запропонованих проектів чи претендентів, входи яких явно недетерміновані, представлені невизначеними змінними, а виходом є вектор двійкового типу (1/0).

У прикладі мова йде про вибір проектів із 8 запропонованих, для кожного задано: бажана інвестиція (в певних межах), шанс на успіх і очікувана віддача, обмеження — фіксований бюджет (ПЧ).

Симулятор серією спроб (рис. 8) формує набори значень входів (p1 ... p8) за визначеними розподілами, статистичними функціями обчислюються узагальнені показники, які надалі використовує оптимізаційна модель, формуючи варіанти виходів.

i	p1					p2					p8				
	d	e	f	g	h	d	e	f	g	h	d	e	f	g	h
1	400000	1	400000	200000	200000	400000	1	400000	300000	100000	400000	1	400000	300000	100000
2	400000	1	400000	100000	300000	400000	1	400000	200000	200000	400000	1	400000	100000	300000
3	400000	1	400000	200000	200000	400000	1	400000	200000	200000	400000	0	0	100000	-100000
4	400000	0	0	200000	-200000	400000	1	400000	100000	300000	400000	1	400000	300000	100000
5	400000	1	400000	100000	300000	400000	0	0	200000	-200000	400000	1	400000	200000	200000
6	500000	1	500000	300000	200000	500000	0	0	200000	-200000	500000	1	500000	200000	300000
7	500000	0	0	200000	-200000	500000	1	500000	100000	400000	500000	0	0	200000	-200000
8	400000	1	400000	300000	100000	400000	0	0	200000	-200000	400000	0	0	200000	-200000
9	400000	1	400000	200000	200000	400000	1	400000	200000	200000	400000	1	400000	100000	300000
10	400000	0	0	200000	-200000	400000	0	0	300000	-300000	400000	1	400000	200000	200000
11	400000	1	400000	300000	100000	400000	1	400000	200000	200000	400000	0	0	100000	-100000
12	500000	0	0	200000	-200000	500000	0	0	100000	-100000	500000	0	0	100000	-100000
13	400000	1	400000	300000	100000	400000	1	400000	100000	300000	400000	0	0	100000	-100000
14	500000	1	500000	300000	200000	500000	1	500000	200000	300000	500000	1	500000	200000	300000
15	400000	1	400000	300000	100000	400000	1	400000	100000	300000	400000	1	400000	100000	300000
16	500000	0	0	300000	-300000	500000	1	500000	100000	400000	500000	0	0	100000	-100000
17	500000	0	0	100000	-100000	500000	1	500000	100000	400000	500000	1	500000	300000	200000
18	400000	1	400000	200000	200000	400000	1	400000	200000	200000	400000	0	0	300000	-300000
19	400000	1	400000	300000	100000	400000	0	0	200000	-200000	400000	0	0	100000	-100000
20	400000	0	0	100000	-100000	400000	1	400000	100000	300000	400000	0	0	100000	-100000
430000	1,0	270000	220000	50000	430000	1,0	300000	170000	130000	130000	430000	0,5	215000	170000	45000
План															
Пр1	Пр2	Пр3	Пр4	Пр5	Пр6	Пр7	Пр8	ЛЧ	ПЧ						
1	1	0	1	0	0	1	0	720000,0	1500000						
								295000,0	ЦФ						

Рис. 8. Фрагмент робочої таблиці

Висновки

Стандартні засоби Excel дають змогу реалізувати ІМ нескладної задачі навчального характеру, аби зрозуміти можливості та принципи побудови моделей цього класу, не звертаючись до спеціальних надбудов. При цьому треба враховувати, що на цьому інструментальному рівні (засоби Excel) існує компроміс між розміром задачі (числом невизначених змінних) і простотою моделі з одного боку та машинними ресурсами і тривалістю її реалізації з іншого, який прямо пов'язаний з роботою ГВЧ. Бо специфікою ІМ та SO-технології у середовищі електронних таблиць необхідною є додаткова таблиця з формулами із сотнями рядків і кількістю стовпців у вигляді суми входів, проміжних та основних виходів. Скажімо, у простому прикладі 3 до основної додається таблиця розміром 300×25 , а в 4-му — додається таблиця з ГВЧ розміром $n \times 5m$, де n — число спроб, m — число шуканих змінних.

1. Кузьмичов А.І. та ін. Ймовірнісне та статистичне моделювання в Excel для прийняття рішень: навч. посіб. — Київ: Вид-во Ліра-К, 2019. 200 с.
2. Camm J., et al. Business Analytics, 3-ed. Cengage Learn., 2019. 818 p.

3. Ragsdale C. Spreadsheet Modeling and Decision Analysis. A Practical Introduction to Business Analytics, 8-ed. Cengage Learn., 2018. 869 p.
4. Evans J. Business Analytics. Methods, Models, and Decision, 2-ed. – Pearson, 2017. 653 p.
5. Powell S., Baker K. Business Analytics. The Art of Modeling with Spreadsheets, 5-ed. Wiley, 2017. 555 p.
6. Кузьмичов А.І. Аналіз ризиків у мережевих моделях проектів засобами імітаційної оптимізації. *Реєстрація, зберігання і оброб. даних*. 2017. Т. 19. № 1. С. 72–80. <https://doi.org/10.35681/1560-9189.2017.19.1.126496>
7. Додонов О.Г., Кузьмичов А.І. Моделювання ризиків у проектно-орієнтованому організаційному управлінні засобами стохастичної оптимізації середовища ASP: сб. трудов конференции «Моделирование-2016», ШМЕ НАНУ, 2016. Т. 1. С. 187–190.
8. Najjar A. Practical Monte Carlo Simulation with Excel. Part 1. Basics and Standard Procedures, 193 p. Part 2. Application and Distributions. XinXii, 2016. 436 p.
9. Кузьмичов А.І. Планування та управління проектами. Моделювання засобами MS Excel. Практикум. Київ: Видавництво Ліра-К, 2016. 180 с.
10. Cox L. Breakthroughs in Decision Science and Risk Analysis. Wiley, 2015. 331 p.
11. ДСТУ ІЕС/ISO 31010:2013. Керування ризиком. Методи загального оцінювання ризику. Київ: МЕРТ України, 2015. 80 с.
12. Law A. Simulation Modeling and Analysis, 5-ed. McGraw Ed., 2013. 800 p. (Пер. рус. А. Лоу, Д. Кельтон. Имитационное моделирование, 3-изд. Санкт-Петербург: Питер: BHV, 2004. 847 с.).

Надійшла до редакції 20.07.2020